/**
 * Copyright (c) 2010, iDA MediaFoundry
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * - Redistributions of source code must retain the above copyright notice,
 *   this list of conditions and the following disclaimer.
 *
 * - Redistributions in binary form must reproduce the above copyright notice,
 *   this list of conditions and the following disclaimer in the documentation
 *   and/or other materials provided with the distribution.
 *
 * - Neither the name of iDA MediaFoundry nor the names of its contributors may
 *   be used to endorse or promote products derived from this software without
 *   specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */
package be.idamf.flash.exec
{
	/**
	 * Pseudo-thread implementation.
	 * 
	 * @author Jan Van Coppenolle
	 * 
	 * @param work Callback function that takes no arguments and return a
	 * Boolean true or false wether or not the job has finished.
	 * @param timeFrame Number of ms to max spend on a job in one frame.
	 */
	public function pseudoThreadExec(work:Function, timeFrame:uint = 0):void
	{
		new PseudoThreadExecution(work, timeFrame);
	}
}

import flash.display.Sprite;
import flash.events.Event;
import flash.utils.Dictionary;
import flash.utils.getTimer;

internal class PseudoThreadExecution
{
	private static const EXECUTIONER:Sprite = new Sprite();
	private static const THREAD:Dictionary = new Dictionary(true);

	public var work:Function;
	public var timeFrame:uint;

	public function PseudoThreadExecution(work:Function, timeFrame:uint)
	{
		this.work = work;
		this.timeFrame = timeFrame;

		if (THREAD[work])
		{
			EXECUTIONER.removeEventListener(Event.ENTER_FRAME, THREAD[work]);
		}

		EXECUTIONER.addEventListener(Event.ENTER_FRAME, executionHandler);
		THREAD[work] = executionHandler;
	}

	public function executionHandler(event:Event):void
	{
		var startTime:int = getTimer();
		var hasWork:Boolean;

		do
		{
			hasWork = work();
		} while (hasWork && getTimer() - startTime < timeFrame);

		if (!hasWork)
		{
			EXECUTIONER.removeEventListener(Event.ENTER_FRAME, executionHandler);
			delete THREAD[work];
		}
	}
}